# EGY KOMPUTER ALGEBRAI PROGRAM A MATEMATIKA OKTATÁSÁBAN

ZIBOLEN ERZSÉBET

***ZIBOLEN Erzsébet*** *adjunktus, Matematika-statisztika tanszék.*

*Végzettsége: matematika-fizika ábrázoló geometria szakos középiskolai tanár és közlekedésmatematikai szakmérnök.*

*Főiskolánkon 4 éve tanít, elsősorban statisztikát, továbbá informatikát és matematikát. Főbb kutatási területe a matematika-statisztika oktatás korszerűsítése, a számítógépes programok bekapcsolása az oktatásba. Több mint 20 éve érdeklődik és publikál a gazdasági matematika nem lineáris problémáiról és ezek számítógépes megoldásairól.*

\+ + + + +

Az utóbbi években a felsőoktatásban a matematika oktatásában és alkalmazásában világszerte egyre inkább elterjedőben vannak az úgynevezett komputer algebrai programok, a legjobbak egyikét Maplenak hívják és hazánkban több egyetemen, de már egyes főiskolákon is használják szimbolikus és numerikus számítások végzésére, ábrázolásra, stb. Ez az írás megkísérli leírni a szóban forgó, DOS-, Windows alapú PC, illetve Macintosh számítógépeken futó, nem nagy hardverigényű szoftver legfontosabb jellemzőit, úgy, hogy bemutatja, hogyan lehet vele feladatokat megoldani a főiskolánkon korábban és jelenleg tanított következő matematikai területekről: analízis (pl. sorozatok határértéke, függvények határértéke, differenciálása, vizsgálata, integrálása és ábrázolása), lineáris algebra (pl. mátrixműveletek, invertálás, rang és determináns, lineáris egyenletrendszerek megoldása), operációkutatás (elemi bázis-transzformáció, lineáris programozási feladatok és megoldásuk), statisztika (pl. leíró statisztika: középértékek, szóródási mutatók, statisztikai ábrafajták, hisztogrammok, korreláció és regresszió).

A továbbiakban a főiskolánkon használt következő tankönyvekre fogunk hivatkozni.

[I.] Analízis. Szerkesztette Dr. Csernyák László. Tankönyvkiadó, 1991.

[II.] Analízis feladatgyűjtemény. Bognár Endre-Fejes Ferenc. Külkereskedelmi Főiskola, 1992.

[III.] Operációkutatás I. Szerkesztette Dr. Tóth Irén. Tankönyvkiadó, 1991.

[IV.] Feladatgyűjtemény a lineáris algebrához. Imre Klára. Külkereskedelmi Főiskola, 1991.

[V.] Operációkutatás II. Szerkesztette Dr. Csernyák László. Tankönyvkiadó, 1990.

[VI.] Alkalmazott statisztika I. Horváth Gézáné dr. Külkereskedelmi Főiskola, 1994.

[VII.] Statisztikai módszerek a gazdasági elemzéshez, I. és II. Kerékgyártó Györgyné-Mundruczó György. AULA Kiadó, 1994.

Az alábbiakban ízelítőül közlünk egy-két matematika feladatot végeredményével együtt, és megadjuk a problémát megoldó Maple parancsokat.

1.) $\lim\limits_{x\to 0}\dfrac{e^x-1}{x}=1$. ([I.], Csernyák: Analízis, 104.o.)

2.) $\left(\sin^3(2x-1)\right)'=6\sin^2(2x-1)\cos(2x-1)$ ([I.], 143.o.)

3.) Az $\begin{matrix} ax+by=1 \\ 2cx+dy=3 \end{matrix}$ egyenletrendszer megoldása: $x=\dfrac{d-3b}{da-2bc}$, $y=\dfrac{-2c+3a}{da-2bc}$.

4.) Az $x_1+3x_2+2x_3$ célfüggvénynek a $2x_1+x_2-x_3\le 10$, $x_1+x_2\le 6$, $-x_2+2x_3\le 16$, $x_1,x_2,x_3\ge 0$ feltételekhez tartozó maximuma az $x_1=0, x_2=6, x_3=11$ pontban következik be. ([V.], Csernyák: Operációkutatás II, 42. o.)

5.) Ha

$x_1=67, x_2=76, x_3=81, x_4=72, x_5=76, x_6=83, x_7=85, x_8=91, x_9=81, x_{10}=88$ és $y_1=6.1, y_2=6.4, y_3=6.7, y_4=6.7, y_5=6.9, y_6=7.1, y_7=7.4, y_8=7.5, y_9=7.6, y_{10}=7.6$, akkor az $x$ magyarázó és az $y$ függő változó közötti lineáris kapcsolatot leíró, a legkisebb négyzetek módszerével megkapható regressziós függvény: $\hat{y}=2{,}09465+0{,}06132x$. ([VII.], Kerékgyártóné: Statisztika II, 349.o.)



Magyarázat: A > jel a Maple promptja, ezután kell beírni a Maple parancsokat, amelyeket pontosvesszővel vagy kettősponttal kell lezárni. Enter hatására a parancsok végrehajtásra kerülnek és ha pontosvesszővel zártuk le őket, az eredmény képernyőre is kerül. Az egyszerre végrehajtott utasításokat kívánságunk szerint a Maple vízszintes vonallal jelöli meg.

A példákból láthatjuk, hogy a Maple rendkívül hatékony munkára képes.

Hogy mi a Maple? A Maple komputer algebrai programot több egyetem kutatói és diákjai fejlesztették ki, elsősorban a Kanadai Waterloo-, Drexel- a Svájci Szövetségi Egyetemen, illetve a Waterloo Maple Software cégnél. A Maple képes sokféle szimbolikus (formális) művelet végzésére az elemi matematika, az analízis, lineáris algebra területein és sok más matematikai részterületen is.

### *A Maple az analízisben*

Most néhány példát szeretnénk megoldani az analízis területéről. Feltételezzük, hogy elindítottuk a Maple program Windowsos változatát. A > jel mögötti Maple utasítást (input) mi írjuk be (ez ahogy szokás, Ariel CE betűtípussal félkövéren van szedve) és általában megadjuk a Maple-számítás eredményét (output) Courier New CE betűvel szedve. Az utasítások mögött sokszor látható # jel a megjegyzés jele, ezt és a mögötte álló karaktereket a program nem tekinti hibának, a végrehajtás során egyszerűen figyelmen kívül hagyja.

Példa: Számítsuk ki a $\pi/12$ tangensét!

**> valos_szam:=tan(Pi/12)^2; # tan a tangens függvény helyett áll;**

$$valos\_szam := \left(2-\sqrt{3}\right)^2$$

Tehát a Maple szerint - és igaza van - $\mathrm{tg}(\pi/12)$ értéke $(2-\sqrt{3})^2$.

A négyzetre emelés (ebben a szimbolikus gyökös formában) elvégezhető:

**> expand(valos_szam); # A szimbolikus eredmény azonosan átalakítható**

$$7-4\sqrt{3}$$

* Nagyszámú numerikus függvény található a Maple-ben, amely tetszőleges számú jegyből álló számra tetszőleges számítási pontossággal alkalmazható:

Példa. Számítsuk ki az előző kifejezést például 25 jegy pontossággal:

**> evalf(valos_szam,25);**

$$.0717967697244908258902148\hat{2}$$

* Függvények illetve adatok két- és háromdimenziós ábrákon szemléltethetőek:

Példa ([I.], Csernyák: Analízis, 187.o. 6.11. pl.):

Ábrázoljuk az $x^3/3 - 3x^2 + 8x$ függvényt!

```
> plot(x^3/3-3*x^2+8*x,x=-1/4..5); # Az ábrát a -1/4, 5 intervallumon készítjük el
```




Példa. Ábrázoljuk az $\dfrac{1}{\left(x^2+y^2\right)\left((x+y)^2+y^2\right)}$ egyenletű felületet! (A megoldásnál a felület x= -3, x=1, y= -2, y=2, z=0, z=4 téglatestbe eső részére szorítkozunk.)

```
> plot3d(1 /(x^2+y^2)/((x+2)^2+y^2), x=-3..1, y=-2..2, view=0..4 );
```



Példa ([II.], Bognár-Fejes: Analízis feladatgyűjtemény, 64. o. 5.97.e) pl.)

Ábrázoljuk a $z = 4x^2 + 3y^2$ egyenletű felületet és szintvonalatait!

```
> plot3d(4*x^2+3*y^2, x=-2..2, y=-2..2, style=CONTOUR, scaling=CONSTRAINED,
> orientation=[0,0], axes=NORMAL); # style=CONTOUR adja a szintgörbéket
```



* A Mapleben sok, műveleteket és algoritmusokat tartalmazó szubrutincsomag található olyan különböző matematikai területekről, mint az elemi matematika, analízis, geometria, statisztika, diszkrét matematika, stb.

Példa ([I.], Csernyák: Analízis, 176.o. 6.4. pl.)

Határozzuk meg az `f(x) = 6x / (1 + x²)` függvény lokális szélsőértékeit és szélsőértékhelyeit!

**> readlib(extrema); # A szélsõértékszámító (extrema) algoritmus beolvasása**

```
proc(fcn,cnstrnts,vars,candidates) ... end
```

**> f:=6*x/(1+x^2); # A függvény elõírása**

$$6 \frac{x}{1 + x^2}$$

**> extrema(f,{},x,szelsoertek_helyek); # A szélsõértékek az alábbiak:**

$$\{-3, 3\}$$

**> szelsoertek_helyek; # A megfelelõ szélsõértékhelyek:**

$$\{\{x = -1\}, \{x = 1\}\}$$

* Speciális, ún. „csőrajz" készítése a Maple-ben. Az első lépés a plots szubrutingyűjtemény beolvasása kell, hogy legyen. A plots az eddigieken kívül sokféle speciális ábrázolási lehetőséget kínál, az animációtól (animate) kezdve a csőrajzzal (tubeplot) bezárólag.

Álljon itt egy körvonalat (paraméteres egyenlete: x=10*cos(t), y=10*sin(t), z=0, 0<=t<=2*3.14) és egy önmagába záródó csavarvonalat (x=cos(t)*(10+4*sin(9*t)), y=sin(t)*(10+4*sin(9*t)), z=4*cos(9*t), 0<=t<=2*3.14) mint középvonalat csővel körülvevő "csőrajzhoz" vezető utasítássorozat és maga a rajz:

**> with(plots): # A "csõrajz"-hoz be kell olvasni a speciális rajzrutinokat**

> R:=8:

> tubeplot({[10*cos(t), 10*sin(t), 0, t=0..2*Pi, # 1. középvonal egyenlete

> radius=2, numpoints=10*R, tubepoints=2*R],

> [cos(t)*(10+4*sin(9*t)),sin(t)*(10+4*sin(9*t)),4*cos(9*t),t=0..2*Pi, # 2.középvonal

> radius=1, numpoints=trunc(37.5*R), tubepoints=R]}, scaling=UNCONSTRAINED,

> orientation=[76,34], shading=NONE, light=[70,70,1,.9,.75]);



* A Maplenek van egy hatékony programnyelve, amellyel saját céljainkra programot írhatunk.

Példa: Tegyük fel, hogy $f(x) = 3x^4 + x^3 + 2x^2 + 5x - 4$ elsõ, második, ..., harmadik deriváltját akarjuk kiszámítani:

Írjunk erre parancssoros programot:

**> fv:=3*x^4+x^3+2*x^2+5*x-4; for n from 1 to 3 do print(diff(fv,x$n)); od;**

$$fv := 3x^4 + x^3 + 2x^2 + 5x - 4$$

$$12x^3 + 3x^2 + 4x + 5$$

$$36x^2 + 6x + 4$$

$$72x + 6$$

* A Maple legtöbb matematikai függvényének forráskódjához hozzáférhetünk és módosíthat-juk is azt.

* Sok algebrai és analízis feladat lépésenként történõ megoldásában a hallgatókat speciális eljárások segítik. Ezeket a student nevû szubrutincsomag tartalmazza:

Példa. Számítsuk ki $(\cos(x) + 1)^3 * \sin(x)$ határozatlan integrálját a student csomag nélkül is, de vele is!

1. A student nélküli számítás (közvetlenül a végeredményt adja):

> int((cos(x)+1)^3*sin(x),x);

$$-\frac{1}{4}(\cos(x)+1)^4$$

2. A student csomag helyettesítéses integrálást támogató changevar rutinjával történõ megoldás következik. Ezzel a hallgató saját kézzel történõ integrálását lépésenként ellenõrizni tudja, és esetleges hibáját is megtalálhatja, így azt ki is javíthatja. Kiderülhet, hogy ha számítási hibát vétett, az hol történt, de az is, ha integrálási módszert kell váltania. A következõ elsõ lépésben arra kérjük a Maple-t, hogy a szóban forgó határozatlan integrálban végezze el a $\cos(x)+1=u$ helyettesítést:

**> student[changevar](cos(x)+1=u, Int ((cos(x)+1)^3*sin(x), x), u);**

$$\int -u^3\,du$$

**> # Megj.: egy lépés eredményére a következõ lépésben "-el hivatkozni lehet**

**> value("); # A value utasítás elvégzi az elõzõ, kijelölt mûveletet (itt integrálást)**

$$-\frac{1}{4}u^4$$

**> subs(u=cos(x)+1,"); # Az eredményben visszatérünk az eredeti változóra**

$$-\frac{1}{4}(\cos(x)+1)^4$$

(A **subs(u=cos(x)+1,")** utasítás u=cos(x)+1 kifejezését behelyettesíti a $-\frac{1}{4}u^4$ kifejezésbe.)

Oldjunk meg néhány röviden kiszámolható feladatot!

Példa. ([I.], 149.o. 5.20. pl.) Mi lesz az *ln(x)* függvény negyedik deriváltja az *e* helyen?

**> diff(ln(x), x$4);**

$$-6\frac{1}{x^4}$$

**> subs(x=e,");**

$$-6\frac{1}{e^4}$$

Példa. ([I.], 153.o. 5.23. pl.) Oldjuk meg az $x^5+x-0,5=0$ egyenletet közelítõleg!

**> fsolve(x^5+x-0.5,x);**

$$.4756527435$$

Példa. ([I.], 162.o. 5.30. pl.) Megkeresendő az $e^x$ fv. 1 körüli ötödik Taylor-polinomja!

> **taylor(exp(x), x=1, 5);**

$$e + e(x-1) + \frac{1}{2}e(x-1)^2 + \frac{1}{6}e(x-1)^3 + \frac{1}{24}e(x-1)^4 + O\left((x-1)^5\right)$$

### *Lineáris programozási feladatok megoldása*

> **with(simplex): # A szimplex módszer programjának beolvasása**

Példa. ([V.], Csernyák, 30.o. 1.6. pl.) Határozzuk meg a $600x_1 + 500x_2$ célfüggvénynek a $15x_1 + 20x_2 \le 1440$, $3x_1 + 2x_2 \le 240$, $x_2 \le 50$ és $x_i \ge 0$ $(i = 1,2,3)$ feltételekhez tartozó maximumát! (A bevezetőben már láttuk, hogyan lehet hasonló feladatot két sorban megoldani, most olyan megoldást adunk, amelyben új, beszédes nevű változókat használunk, illetve megmutatjuk, hogy az angol nyelvű Maple parancsokat (itt a maximize parancsot) is lehet magyarítani.

> **maximalizalj:=maximize; # Az LP-t megoldó parancs magyarítása**

$$maximalizalj := maximize$$

> **NEMNEGATIV:=NONNEGATIVE; # A nemnegativitási feltétel magyarítása**

Vezessük be most a korlátozó feltételekre a korlatozasok nevű, a célfüggvényre pedig a celfuggveny nevű változókat!

> **korlatozasok:={15*x1+20*x2<=1440, 3*x1+2*x2<=240, x2<=50};**

$$korlatozasok := \{15x1 + 20x2 \le 1440, 3x1 + 2x2 \le 240, x2 \le 50\}$$

> **celfuggveny:=600*x1+500*x2;**

$$celfuggveny := 600x1 + 500x2$$

> **maximalizalj(celfuggveny,korlatozasok,NEMNEGATIV);**

$$\{x1 = 64, x2 = 24\}$$

Tehát a célfüggvény egy, a korlátozásoknak megfelelő maximumhelye az x1=64, x2=24 pont.

* A lineáris programozási feladat megjeleníthető mátrixalakban:

> **display(korlatozasok); # Feladatunk feltételrendszerének kiírása**

$$\begin{bmatrix} 15 & 20 \\ 3 & 2 \\ 0 & 1 \end{bmatrix} \begin{bmatrix} x1 \\ x2 \end{bmatrix} \le \begin{bmatrix} 1440 \\ 240 \\ 50 \end{bmatrix}$$

* Egy lineáris programozási feladat duálisa könnyen előállítható.

Példa. Határozzuk meg utolsó LP feladatunk duálisát! A duális változót jelölje u!

> dual(celfuggveny, korlatozasok,u); # A következõ sor adja a duális feladatot

$1440u1 + 240u2 + 50u3, \{500 \leq 20u1 + 2u2 + u3, 600 \leq 15u1 + 3u2\}$

* Szállítási feladat megoldható LP-ként, minthogy átfogalmazható lineáris programozási feladatra és a Maple nagyméretű feladatokat is hatékonyan kezel. Például a [V.], 72.o., 2.1. szállítási feladat egyenlőtlenségeit egyenlőségekre cserélve a 12 változós modellt 6 másodperc alatt oldotta meg egy 486-os PC-n.

## *Lineáris algebrai feladatok megoldása*

A bevezetőben említettük, hogy a Maple komputer algebrai program és ehhez méltóan különösen alkalmas lineáris algebrai feladatok megoldására. Remélhetően az olvasónak is ez lesz a véleménye a következő feladatok megoldása láttán.

Példa. ([III.], Tóth Irén, Operációkutatás I, 62.o. 3.1.2. pl.) Transzponáljunk egy A mátrixot!

> with(linalg): # Elsõ lépésként olvassuk be a lineáris algebra szubrutinjait!

> A:=array([[2, 3, 1], [0, 4, 5]]); # A mátrix megadása ún. tömbként

$$A := \begin{bmatrix} 2 & 3 & 1 \\ 0 & 4 & 5 \end{bmatrix}$$

> transpose(A); # A transzponálása, a transzponált a következõ sorban látható

$$\begin{bmatrix} 2 & 0 \\ 3 & 4 \\ 1 & 5 \end{bmatrix}$$

Példa. ([III.], 65.o. 3.2.2. pl.) Adjunk össze két mátrixot!

> A:=array([[2,-3,5,1],[1,0,-2,5]]); # Ez a mátrixok egyfajta megadási módja

$$\begin{bmatrix} 2 & -3 & 5 & 1 \\ 1 & 0 & -2 & 5 \end{bmatrix}$$

> B:=matrix(2,4,[1,3,2,7,-1,-2,0,3]); # Ez a mátrixok másféle megadási módja

$$\begin{bmatrix} 1 & 3 & 2 & 7 \\ -1 & -2 & 0 & 3 \end{bmatrix}$$

> evalm(A+B); # Így lehet az elõbb kapott A és B mátrixokat összeadni

$$\begin{bmatrix} 3 & 0 & 7 & 8 \\ 0 & -2 & -2 & 8 \end{bmatrix}$$

Példa. ([III.], 66. o. 3.2.3. pl.) Szorozzunk mátrixot számmal!

**> A:=array([[2,4],[6,-1],[0,5]]); # Az A mátrix:**

$$\begin{bmatrix} 2 & 4 \\ 6 & -1 \\ 0 & 5 \end{bmatrix}$$

**> evalm(3*A); # Az A mátrix 3-mal van szorozva**

$$\begin{bmatrix} 6 & 12 \\ 18 & -3 \\ 0 & 15 \end{bmatrix}$$

Példa. ([III.], 71. o. 3.2.11. pl.) Vegyük vektorok skaláris szorzatát!

**> dotprod([2,1,3,0,-1],[0,-3,4,5,-2]); # 2*0+1*(-3)+3*4+0*5+(-1)*(-2)=-3+12+2=11**

$$11$$

Példa. ([III.], 76. o. 3.2.18. pl.) Nézzük, hogyan lehet mátrixot mátrixszal szorozni.

**> A:=array(1..3,1..3,[[2,3,4],[1,-1,2],[0,6,5]]):**

**> B:=array(1..3,1..3,[[2,3,4],[1,-1,2],[0,6,5]]): # A , B megadása a kiírást elnyomva**

**> print(A,B); # Így írathatjuk ki egymás mellé az A és B mátrixokat egyszerre**

$$\begin{bmatrix} 2 & 3 & 4 \\ 1 & -1 & 2 \\ 0 & 6 & 5 \end{bmatrix}, \begin{bmatrix} 2 & 3 & 4 \\ 1 & -1 & 2 \\ 0 & 6 & 5 \end{bmatrix}$$

**> evalm(A&*B); # Az A és B mátrixok szorzata**

$$\begin{bmatrix} 7 & 27 & 34 \\ 1 & 16 & 12 \\ 6 & 24 & 37 \end{bmatrix}$$

Példa. ([III.], 138.o. 5.3.1) Oldjuk meg az alábbi egyenletrendszert,

x1+2x2+6x3+3x5=1, -x1-x2-4x3-x4-x5= -2, 2x1+x2+6x3+3x4=5 !

Az ismeretlenek együtthatóinak mátrixát jelölje *A*, a jobboldal vektorát pedig *b*. Ezzel az egyenletrendszer `Ax = b` alakba írható át.

**> A:=array([[1,2,6,0,3],[-1,-1,-4,-1,-1],[2,1,6,3,0]]):b:=array([1,-2,5]):**

**> print(A,b); # A és b visszaírása**

$$\begin{bmatrix} 1 & 2 & 6 & 0 & 3 \\ -1 & -1 & -4 & -1 & -1 \\ 2 & 1 & 6 & 3 & 0 \end{bmatrix}, \begin{bmatrix} 1 & -2 & 5 \end{bmatrix}$$

1. megoldási módszer

**> x:=linsolve(A,b); # Ezzel kapjuk az egyenletrendszer megoldását.**

$$x := \begin{bmatrix} -2\_t_1 + \_t_3 + 3 - 2\_t_2 & -2\_t_1 - 2\_t_3 - 1 + \_t_2 & \_t_1 & \_t_2 & \_t_3 \end{bmatrix}$$

Ellenőrizzük a megoldást az $Ax$ szorzat kiszámításával!

**> evalm(A&*x); # Láthatjuk, hogy valóban a jobboldalt kapjuk vissza:**

$$\begin{bmatrix} 1 & -2 & 5 \end{bmatrix}$$

2. megoldási módszer, Gauss-féle elimináció

**> C:=augment(A,b); # Elkészítjük az úgynevezett kibõvített mátrixot**

$$C := \begin{bmatrix} 1 & 2 & 6 & 0 & 3 & 1 \\ -1 & -1 & -4 & -1 & -1 & -2 \\ 2 & 1 & 6 & 3 & 0 & 5 \end{bmatrix}$$

**> G:=gaussjord(C); # Az egyenletrendszer Gauss-féle alakra transzformálása**

$$G := \begin{bmatrix} 1 & 0 & 2 & 2 & -1 & 3 \\ 0 & 1 & 2 & -1 & 2 & -1 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

**> x:=backsub(G); # Az elõzõ alak szerinti redukált egyenletrendszer megoldása**

$$x := \begin{bmatrix} -2t3 - 2t4 + t5 + 3 & -2t3 + t4 - 2t5 - 1 & t3 & t4 & t5 \end{bmatrix}$$

Ez láthatóan a megoldás egy másik alakja. Természetesen itt is három paraméter van.

Példa. ([III.], 5.4.2. pl.) Adott mátrix inverzének meghatározása

**> A:=array([[1,2,1],[2,5,4],[2,6,7]]);**

$$A := \begin{bmatrix} 1 & 2 & 1 \\ 2 & 5 & 4 \\ 2 & 6 & 7 \end{bmatrix}$$

**> B:=inverse(A); # Az inverz 1. meghatározása az inverse paranccsal**

$$B := \begin{bmatrix} 11 & -8 & 3 \\ -6 & 5 & -2 \\ 2 & -2 & 1 \end{bmatrix}$$

**> B:=evalm(A^(-1)); Az inverz 1. meghatározása mint mínusz egyedik hatvány**

$$B := \begin{bmatrix} 11 & -8 & 3 \\ -6 & 5 & -2 \\ 2 & -2 & 1 \end{bmatrix}$$

Ellenőrizzük, hogy valóban A inverzét kaptuk-e meg B-ben!

**> evalm(A&*B); # Igen, mert A és B szorzata az egységmátrix lett, lásd alant**

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

* Példa. ([III.], 164.o. 5.7.2. pl.) Határozzuk meg egy mátrix sajátértékeit és sajátvektorait!

**> A:=array([[1,1],[4,1]]);**

$$A = \begin{matrix} 1 & 1 \\ 4 & 1 \end{matrix}$$

**> eigenvals(A); # A sajátértékek:**

$$3,\ -1$$

**> eigenvects(A); # A sajátvektorok és hozzátartozó sajátértékeik:**

$$[-1,\ 1,\ \{[1\quad -2]\}],\ [3,\ 1,\ \{[1\quad 2]\}$$

Ennek az eredménynek az értelmezése a következő: -1 egyszeres sajátérték és a hozzá-tartozó sajátvektor [1, -2], ill. 3 egyszeres sajátérték és a hozzátartozó sajátvektor [1, 2].

Példa. ([IV.] Imre Klára, Feladatgyűjtemény a lineáris algebrához, 55.o. 134. pl.)

**> A:=array([[2,3,5],[1,5,6]]); # A feladat A rangjának meghatározása**

**> rank(A); # Az A mátrix rangjára a rank paranccsal 2 adódik**

$$2$$

* Hasonló könnyedséggel kaphatjuk meg például egy **A** mátrix determinánsát **>det(A)**, hatványait (például köbét) **>evalm(A^3)** alakban, és így vizsgálhatjuk meg egy mátrix projektor voltát és nilpotenciálosságát, stb.

***Statisztikai feladatok megoldása a Maple-ben***

Példa. ([VI.], Horváth Gézáné dr., Alkalmazott statisztika I. 23.o. 2.6. pl.) Osszuk be a 18, 20, 25, 18, ..., 47 (összesen száz elemű) mintát 5 hosszúságú intervallumokba!

**> with(stats): # A statisztikai programcsomag beolvasása**

**> adatsor:=[18, 20, 25, 30, 37, 18, 22, 27, 32, 55, 60, 32, 35, 45, 47, 51, 18, 23, 37, > 42, 57, 62, 75, 67, 65, 22, 27, 32, 32, 45, 27, 26, 25, 41, 42,**

**52, 53, 35, 40, 50, 19, > 20, 21, 30, 25, 30, 38, 42, 45, 47]; # A mintát átadjuk az adatsor nevû listának**

```
adatsor:=[18,20,25,30,37,18,22,27,32,55,60,32,35,45,47
,51,
18,23,37,42,57,62,75,67,65,22,27,32,32,45,27,26,25,41,
42,
52,53,35,40,50,19,20,21,30,25,30,38,42,45,47]
```

2. lépés. Az adatsor adatait most soroljuk be a [15,20), [20,25), ..., [65,1000) osztályokba:

**>transform[tallyinto](adatsor,[15..20,20..25,25..30,30..35,35..40,40..45, 45..50,**
**>50..55,55..60,60..65,65..100]): # Az adatok szétbontása**

**> transform[statsort]("); # Sorba rendezzük az osztályokat**

```
[Weight(15..20,4),Weight(20..25,6),Weight(25..30,7),
 Weight(30..35,7),Weight(35..40,5),Weight(40..45,5),
Weight(45..50,5),Weight(50..55,4),Weight(55..60,2),
Weight(60..65,2),Weight(65..100,3)]
```

Az eredmény úgy értelmezhető, hogy 4 olyan adat van amely 15-nél kisebb vagy egyenlő és 20-nál kisebb, 6 olyan adat van, amely 20-nál nagyobb vagy egyenlő és 25-nél kisebb, stb.

* Készítsük el az előző adatsor hisztogramját!

**> statplots[histogram](adatsor); # Az adatsor hisztogramja:**




Példa.Oldjuk meg most [VI.], 50.o. 4.1. feladatát! Az a kérdés, hogy ha a 600-799 osztályba 3, a 800-999 osztályba 7, ... , végül a 2000-2199 osztályba 4 adat esik, akkor mekkora az adatok átlagos értéke.

**> adatok:=[Weight(600..799, 3), Weight(800..999, 7), Weight(1000..1199, 11),**

>Weight(1200..1399, 22), Weight(1400..1599, 40), Weight(1600.. 1799,24),

>Weight(1800..1999, 9), Weight(2000..2199, 4)]; # Az adatok megadása

```
adatok:=[Weight(600..799,3), Weight(800..999,7), Weight(1000..1199,11),Weight(1200..1399,22),Weight(1400.. 1599,40,Weight(1600..1799,24),Weight(1800..1999,9),Weight (2000..2199,4)]
```

A keresett átlag (nyilván aritmetikai):

**> describe[mean](adatok); # Az aritmetikai átlagnak a mean parancs felel meg.**

---

> evalf("); # Az elõzõ átlag numerikus értéke

```
1461.166667
```

* Számítsuk ki az előző adatok harmonikus, geometriai, négyzetes átlagát! A megfelelő parancsok neve harmonicmean, geometricmean, quadraticmean lesz.

>h:=describe[harmonicmean](adatok): g:=describe[geometricmean](adatok):

>k:=describe[quadraticmean](adatok): print(evalf(h), evalf(g), evalf(k));

```
1388.154995, 1426.986430, 1491.488714
```

A Maple alkalmas adatlisták szokásos helyzeti középértékeinek vagyis móduszának, kvantiliseinek (mediánjának, kvartiliseinek, deciliseinek) kiszámítására, illetve utóbbiakat speciális ábrán szemléltetni is tudja.

> describe[median](adatok); # A medián (l. még [VI.], 57.o. példája)

```
1487
```

> evalf(describe[mode](adatok)); # A módusz (l. még [VI.], 57.o.)

```
1505.382353
```

> evalf(describe[quartile[1]](adatok)); # Az elsõ kvartilis (l. még [VI.], 56.o. 4.5.)

```
1281.318182
```

> evalf(describe[quartile[2]](adatok)); # A második kvartilis (vagyis a medián)

```
1484.500000
```

Példa. ([VI.], 64.o. 5.1. pl.) Számítsuk ki a szóródás néhány mutatóját!

Vegyük elsőnek az átlagos abszolút eltérést!

```
> adatok:=[Weight(15..25, 10), Weight(25..35, 14), Weight(35..45, 10),
> Weight(45..55, 9), Weight(55..65, 4), Weight(65..75, 3)]; # I. [A], 5.1. pl.
```

```
adatok:=[Weight(15..25, 10), Weight(25..35, 14), Weight(35..45, 10), Weight(45..55, 9), Weight(55..65, 4), Weight(65..75, 3)]
```

```
> describe[meandeviation](adatok); # Ez az átlagos abszolút eltérés
```

```
> describe[standarddeviation](adatok); # A szórás
```

```
> describe[variance](adatok); # A szórásnégyzet
```

```
> describe[coefficientofvariation](adatok); # Relatív szórás (nem %-os alak)
```

```
> describe[range](adatok); # A szóródás terjedelme
```

```
15..75
```

* Példa. ([VII.], Kerékgyártóné, II. 349.o.) Írjuk fel a regressziós egyenes egyenletét, ha *x* értékei 67, 76, ..., 88 illetve *y* értékei 6.1, 6.4, ..., 7.6!

```
>xad:=[67,76,81,72,76,83,85,91,81,88];
yad:=[6.1,6.4,6.7,6.7,6.9,7.1,7.4,7.5,7.6,7.6];
```

```
xad:=[67,76,81,72,76,83,85,91,81,88]
yad:=[6.1,6.4,6.7,6.7,6.9,7.1,7.4,7.5,7.6,7.6]
```

```
>regresszios_egyenlet:=fit[leastsquare[[x,y]]]([xad,yad]);
```

```
regresszios_egyenlet := y = 2.094650206+0.6131687242x
```

* Összetettebb feladatokat is megoldhatnánk, ilyen lehetne [VII.], II. 356.o. 8.2. példája, amelyben adott x , y értékpárokhoz ki kell számítani a regressziós egyenest, az y becsült értékeket, valamint a becslés jóságára jellemző alapsokasági szórást!

Példa. ([VII.], 469. o. 10.7. pl.) Parabola trendfüggvény számítandó, ha *x* értékei: 1, 2, ..., 6 , *y* értékei: 17915, ..., 16963

```
> fit[leastsquare[[t,y], y=a+b*t+c*t^2]]
```

```
> ([[1,2,3,4,5,6],[17915,20605,23890,21213,21364,16963]]): evalf(");
```

$$y = 12842.50000 + 5851.446429t - 856.9821429t^2$$

Befejezésként szeretnénk megemlíteni, hogy a Maple programnak Magyarországon is van egy rendkívül kedvező oktatási változata és 1996. februárjában 2000 példányban megjelent az első magyar nyelvű, a Maple-t ismertető, példákkal és ábrákkal gazdagon illusztrált könyv. Adatai: Molnárka-Gergó-Wettl-Horváth-Kallós, "A Maple V és alkalmazásai", Springer Hungarica Kft.